
ABSTRACT  

The Statistical Graphics (SG) procedures and the Graph Template Language (GTL) are capable of 
generating powerful individual data displays.  What if one wanted to visualize how a distribution changes 
with different parameters, how data evolve over time, or to view multiple aspects of a three dimensional 

plot?  By utilizing macros to generate each a graph for each frame, combined with the ODS PRINTER 
destination, it is possible to create .gif files to create effective animated data displays.  This paper outlines 
the syntax and strategy necessary to generate these displays, as well as provides a handful of examples.  

Intermediate knowledge of PROC SGPLOT, PROC TEMPLATE, and the SAS® MACRO language is 
assumed. 

INTRODUCTION  

Using either the SG procedures or GTL in combination with the ODS PRINTER destination can generate 

animated displays of data.  The steps, which will be expanded upon later, are as follows: 

 Determine the general shape of the graph, which will be extended using a macro.  The same 

principle applies to the first step in writing any macro, test one case first.   

 Determine what the macro parameter(s) will be.  The macro will be used to generate one graph 

per value of said parameter.  This will be a value that changes, thereby changing each graph and 

creating each frame used in the animation. 

 Write the macro.  The individual macro written will depend on the type of graph that is desired 

and could require a separate graph per value of the macro parameter, or additionally require a 

separate data set per value of the macro parameter to generate each graph as well.  

 Set the animation options. 

 Invoke ODS PRINTER. 

 Call the macro. 

 Stop the animation. 

 Close ODS PRINTER. 

Writing the macro for the animation, to reiterate, is typically dependent on the type of graphics that are to 
be displayed, and thus general ideas will be expressed in the examples to follow.  Firs t, the general 

syntax will be discussed. 

GENERAL SYNTAX 

Below is the general syntax to create an animated .gif file from SAS® graphics:  

   options printerpath=gif animation=start animduration=<number>   

   animloop=<yes/no> noanimoverlay nodate nonumber; 

 

   ods printer file="<filename>"; 

 

   <call macro to generate graphs> 

   options printerpath=gif animation=stop; 

 

   ods printer close; 

 

The options used are explained as follows: 

 printerpath=gif 

Tells SAS what type of file to put all of the graphs into.  .gif is recommended. 

 animation=start 

Tells SAS to start the animation. 

 animduration= 

Sets frame speed in seconds.  This can be changed to speed up or slow down the animation, 
accordingly. 

 animloop=  

Loop the animation. 

 noanimoverlay 

Replace each graph with the next instead of overlaying. 

 animoverlay 

Overlapy each graph with the next. 

 nodate nonumber 

Suppress the date and page number of each graph. 

 animation=stop 

Tells SAS to stop the animation 

 Note that the syntax for ODS PRINTER is the same for other ODS destinations. 

EXAMPLE 1 – RELATIONSHIP BETWEEN TWO SERIES GRAPHS 

This first example illustrates the basic process of creating an animated .gif with PROC SGPLOT.  
Consider the following SAS data set below (created by generating random numbers) containing one 

independent variable (XVAR) and two dependent variables (Y1VAR, Y2VAR):  

 

Suppose we want to view the relationship between Y1VAR and Y2VAR as XVAR is increasing as an 
animation.  This involves creating a data set starting with the first observation being complete and the rest 
of the observations having missing values of Y1VAR and Y2VAR, and then filling in a value one by one 

until the final data set is complete: 

   

  

Figure 1:  Sample frame data sets for series plots 

Graphing each of these data sets produces the following plots used as animation frames: 

 

 
Figure 2:  Individual animation frames for series plots 

Running a macro that produces all of these graphs inside of the ODS PRINTER block will produce the 

animated .gif file that runs each frame.  See the appendix for the code that generates this.  

EXAMPLE 2 – THE BETA DISTRIBUTION 

This example illustrates the changes in the shape of the beta distribution when one parameter is kept 
fixed and the other is allowed to vary.  Consider the following data set where the independent variable 
(XVAR) is generated from from a random uniform distribution and the dependent variable (YVAR) is a 

beta distribution with alpha parameter equal to 2: 

 

In this example, the variable j10 is the variable that defines each frame and is based on the varying 

parameter of the beta distribution, j.  To illustrate the data sets and subsequent graphs generated, 
consider some sample data sets where j = 1, 1.9, and 5: 

   
Figure 3:  Sample frame data sets for the beta distribution 

When fitted to a LOESS curve in PROC SGPLOT, these data sets generate the following plots:  

   

 
Figure 4:  Individual animation frames for the beta distribution 

Note that these are only a sample of 3 graphs; one graph for each value of beta from 1 to 5 by 0.1 
increments are generated for animation.  See the appendix for precise code.  

EXAMPLE 3 – POWER VS. HORIZONTAL SHIFT OF A NORMAL DISTRIBUTION 

This example illustrates what happens to the power of a standard normal distribution test as the mean 
differs from zero.  Consider the following data set that starts the animation.  The anomaly at zero is due to 
the floating point calculation due to looping with a decimal index. 

The variable XVAR is the independent variable, YVAR is the dependent variable that represents the 

standard normal distribution, YVAR2 is the dependent variable that represents the shifted standard 
normal distribution, and LOW and HI are variables that will assist with graphing.  The following data sets 
and subsequent graphs represent the shifted normal distribution moving down the x-axis: 

 

Figure 5:  Sample frame data sets for the shifting normal distribution 

Note the shading that represents an increase in power as the mean shifts from zero:  

 

Figure 6:  Individual animation frames for the shifting normal distribution 

See the appendix for precise code. 

EXAMPLE 4 – ROTATING A THREE DIMENSIONAL PLOT 

When viewing a three dimensional plot, sometimes viewing from only one angle is insufficient.  This 
example illustrates how to rotate a three dimensional plot in order to get a more complete viewing.  It 

utilizes the Graph Template Language and will be different than the previous ones in that the templates 
create the frames instead of individual data sets.  The details of the generation of the graph can be found 
in the SAS® 9.4 Graph Template Language User’s Guide on pages 169-174.  The key to animating this 

plot is to vary the value in the ROTATE= option of the LAYOUT OVERLAY3D statement as the macro 
below outlines: 

   ods escapechar="~"; 

   %macro template; 

   %do i=1 %to 36; 

    /* This creates a macro variable for the number of degrees of rotation */ 

     %let j=%sysevalf(10*&i); 

     proc template; 

     /* Create one template for each rotation */ 

     define statgraph surfaceplot&i; 

       begingraph; 

       entrytitle "SURFACECOLORGRADIENT=TEMPERATURE"; 

     /* Note the ROTATE option that rotates the graph by 10 degrees */ 

         layout overlay3d / cube=false rotate=&j 

        xaxisopts=(display=(line)) 

                            yaxisopts=(display=(line))  

                            zaxisopts=(display=(line)); 

           surfaceplotparm x=length y=width z=depth / name="surf" 

           surfacetype=fill 

           surfacecolorgradient=temperature 

           reversecolormodel=true 

           colormodel=twocoloraltramp; 

           continuouslegend "surf" /  

                         title="Temperature (~{unicode  '00B0'x}F)"; 

         endlayout; 

       endgraph; 

     end; 

     run; 

   %end; 

   %mend; 

   %template; 

After creating the templates, a data set is generated from the LAKE data set in the SASHELP library  as 

outlined in the User’s Guide referenced above.  PROC SGRENDER is then run for each template to 
create the animation frames.  Below are sample graphs: 

   

Figure 7:  Individual animation frames for a rotated three dimensional plot 

CONCLUSION 

While the Statistical Graphics Procedures and the Graph Template Language are powerful tools for 
producing effective data displays, there are times they fall short when showing variation in parameters or 

multiple angles of a three dimensional plot.  The ODS PRINTER destination in combination with these 
tools and a carefully crafted macro comes to the rescue in producing effective and informative 
animations. 

APPENDIX – CODE USED TO GENERATE EACH EXAMPLE 

EXAMPLE 1 

/* Generate the starting data set */ 

data one; 

do XVAR=1 to 10; 

  Y1VAR=ranuni(XVAR); 

  Y2VAR=ranuni(11*XVAR); 

  output; 

end; 

run;  

 

/* Transpose the data in order to easily make each value missing one at a 

time */ 

proc transpose data=one out=two; 

  by XVAR; 

  var Y1VAR Y2VAR; 

run; 

 

/* Create an index variable to assist with missing values */ 

data three; 

  set two; 

  INDEX=_N_; 

run; 

 

 

%macro fill; 

%do j=2 %to 20; 

  /* Create data sets with sequentially missing values */ 

  data three&j; 

  set three; 

  if INDEX > &j then COL1=.; 

  run; 

 

  /* Get the data back into the original structure */ 

  proc transpose data=three&j out=five&j (drop=_NAME_); 

  by XVAR; 

  id _NAME_; 

  var COL1; 

  run; 

   

  /* Create each frame using PROC SGPLOT */ 

  proc sgplot data=five&j noautolegend; 

  series x=XVAR y=Y1VAR / markers markerattrs=(symbol=circlefilled color=red) 

                        lineattrs=(pattern=solid color=red)  

                        legendlabel="Y1 Variable" name="y1"; 

  series x=XVAR y=Y2VAR / y2axis markers markerattrs=(symbol=x color=blue)  

                        lineattrs=(pattern=solid color=blue)  

                        legendlabel="Y2 Variable" name="y2"; 

  xaxis label="X Variable" values=(1 to 10 by 1); 

  yaxis label="Y1 Variable" values=(0 to 1 by 0.2); 

  y2axis label="Y2 Variable" values=(0 to 1 by 0.2); 

  keylegend "y1" "y2"; 

  run; 

%end; 

%mend; 

 

/* Call the macro to generate the animation */ 

options printerpath=gif animation=start animduration=1 animloop=yes 

noanimoverlay nodate nonumber; 

ods printer file="H:\SAS Training\Animation Paper\seriesplot.gif"; 

%fill; 

options printerpath=gif animation=stop; 

ods printer close; 

EXAMPLE 2 

/* Create the starting data set */ 

data beta; 

  do i=1 to 200; 

    do j=1 to 5 by 0.1; 

      XVAR=ranuni(234); 

      YVAR=pdf("beta",XVAR,2,j); 

      output; 

    end; 

  end; 

run; 

/* Create animation variable */ 

data beta2; 

  set beta; 

  j10=round(10*j,1); 

run; 

 

/* The next two blocks of code are used to generate macro variables used in 

the INSET statement in each of the graphs */ 

data beta3; 

  set beta2; 

  if i=1; 

  keep j; 

run; 

proc sql noprint; 

  select j 

  into :BETA10-:BETA50 

  from beta3; 

quit; 

 

ods escapechar="~"; 

%macro betas; 

%do i=10 %to 50; 

  /* Subset on the animation variable */ 

  data beta2&i; 

    set beta2; 

    if j10=&i; 

  run; 

  /* Graph each frame */ 

  proc sgplot data=beta2&i noautolegend; 

    loess x=XVAR y=YVAR/nomarkers; 

    yaxis label="Beta Distribution" values=(0 to 3 by 1); 

    xaxis label="Independent Variable"; 

    inset "~{unicode alpha}=2, ~{unicode beta}=&&BETA&i" / position=topright  

           textattrs=(size=30 weight=bold color=lightgrey); 

  run; 

%end; 

%mend; 

 

/* Call the macro to generate the animation */ 

options printerpath=gif animation=start animduration=0.1 animloop=yes 

noanimoverlay nodate nonumber; 

ods printer file="H:\SAS Training\Animation Paper\beta.gif"; 

%betas; 

options printerpath=gif animation=stop; 

ods printer close; 

EXAMPLE 3 

ods escapechar="~"; 

%macro normals; 

%do i=1 %to 30; 

/* Want to index by 0.1 */ 

%let k=%sysevalf(&i/10); 

/* Shifting the mean point */ 

%let m=%sysevalf(&k + 1); 

/* Create a macro variable for the "scooting" of the shifted distribution */ 

%let j=%sysevalf(&k+4); 

data normal&i; 

  do XVAR=-4 to &j by 0.01; 

    /* Generate values of a standard normal distribution */ 

     YVAR=pdf("normal",XVAR,0,1); 

    /* Generate values of a shifted standard normal distribution */ 

     YVAR2=pdf("normal",XVAR,&m,1); 

    /* LOW is always set to zero as a lower bound for the BAND statement in 

PROC SGPLOT */ 

     LOW=0; 

    /* Shades after the 97.5th percentile */ 

     if XVAR ge 1.96 then HI=YVAR2; 

     else HI=0; 

     output; 

  end; 

run; 

 

/* Generate the plots for each frame */ 

proc sgplot data=normal&i noautolegend; 

  series x=XVAR y=YVAR/lineattrs=(color=black); 

  series x=XVAR y=YVAR2/lineattrs=(color=black); 

  band x=XVAR lower=LOW upper=HI/fillattrs=(color=blue transparency=0.8); 

  /* Create a reference line for the moving delta */ 

  refline &m/axis=x label="~{unicode delta}" labelpos=min; 

  /* Create a reference line at zero */ 

  refline 0/axis=x label="0" labelpos=min; 

  yaxis display=none; 

  xaxis values=(-4 to 6) display=none; 

run; 

%end; 

%mend; 

 

/* Call the macro to generate the animation */ 

options printerpath=gif animation=start animduration=0.1 animloop=yes 

noanimoverlay nodate nonumber; 

ods printer file="H:\SAS Training\Animation Paper\normal slide.gif"; 

%normals; 

options printerpath=gif animation=stop; 

ods printer close; 

CONTACT INFORMATION  

Your comments and questions are valued and encouraged. Contact the author at: 

Jesse Pratt 

Cincinnati Children’s Hospital Medical Center  
513-803-3084  
Jesse.Pratt_2@cchmc.org  

